

Penyelesaian Beragam Persoalan Matematika Terapan dengan Genius Mathematical Tool

Saifuddin Arief
Saifuddin.Arief@rocketmail.com

Lisensi Dokumen:

Copyright © 2003-2019 IlmuKomputer.Com

Seluruh dokumen di IlmuKomputer.Com dapat digunakan, dimodifikasi dan disebarkan secara bebas untuk tujuan bukan komersial (nonprofit), dengan syarat tidak menghapus atau merubah atribut penulis dan pernyataan copyright yang disertakan dalam setiap dokumen. Tidak diperbolehkan melakukan penulisan ulang, kecuali mendapatkan ijin terlebih dahulu dari IlmuKomputer.Com.

Genius merupakan *freeware* untuk komputasi numerik dan visualisasi data. Genius cocok digunakan sebagai sebuah kalkulator biasa maupun untuk menyelesaikan sejumlah persoalan numerik yang muncul dalam dunia rekayasa dan penelitian. Genius hanya dapat dijalankan pada sistem operasi Linux. Informasi dan dokumentasi dan yang terkait dengan Genius dapat dilihat pada situs https://www.jirka.org/genius.html#use.

Pada artikel ini akan diberikan sejumlah penyelesaian dari beberapa persoalan komputasi numerik yang muncul dalam bidang matematika terapan, sains dan rekayasa. Artikel ini merupakan lanjutan dari artikel sebelumnya yaitu Pengenalan Genius Mathematics Tools: Freeware untuk Komputasi Numerik dan Visualisasi Data pada Linux (https://ilmukomputer.org/2018/12/22/pengenalan-genius-mathematics-tools).

Contoh 1.

Sebuah kota kecil dengan penduduk 1000 orang terkena suatu wabah penyakit menular. Jumlah orang yang sakit setelah t hari semenjak menyebarnya wabah penyakit dapat dimodelkan dengan fungsi sebagai berikut:

$$P(t) \, = \, \frac{1000}{\left[1 \, + \, 999 \! \cdot \! e^{(-0.603 \, t)}\right]}$$


Mengacu pada formula di atas, jumlah penduduk yang sakit 10 hari setelah menyebarnya penyakit dapat diestimasi sebagai berikut:

```
genius> P = 1000/(1 + 999*exp(-0.603*10))
= 293.850719578
genius> round(P)
= 294.0
```

Diperoleh jumlah orang yang sakit semenjak 10 hari menyebarnya wabah penyakit adalah 294 orang.



Contoh 2. Berikut ini adalah ilustrasi dari sebuah gaya yang bekerja pada suatu pengungkit.



Gambar 1

Dari ilmu statika, diperoleh besarnya gaya F untuk menghasilkan momen M tersebut adalah

$$F = \frac{M}{\cos(\theta)(a + b\sin(\phi)) - b\sin(\theta)\cos(\phi)}$$

Anggap $\phi = 60^\circ$, $\theta = 30^\circ$, a = 50 mm, serta b = 300 mm maka gaya F yang harus dikerjakan pada ujung pengungkit sehingga menghasilkan momen pada titik O sebesar 15 Nm yang bekerja searah jarum jam dapat ditentukan dengan perhitungan sebagai berikut.

```
genius> M = 15;
genius> phi = 60/180*pi; theta = 30/180*pi;
genius> a = 0.05; b = 0.30;
genius> F = M/(cos(theta)*(a + b*sin(phi)) - b*sin(theta)*cos(phi))
= 77.599076226
```

Jadi gaya yang diperlukan untuk menghasilkan momen sebesar 15 Nm adalah 77.6 N.

Contoh 3.

Apabila pertumbuhan populasi penduduk dianggap bertambah secara eksponensial maka jumlah penduduk pada awal suatu tahun tertentu dapat dihitung dengan formula:

$$y = y_o e^{kt}$$

dimana y_0 adalah jumlah penduduk pada awal tahun acuan, k adalah laju pertambahan penduduk dan tadalah perbedaan waktu terhadap tahun acuan

Misalkan jumlah penduduk dunia pada awal tahun 1990 adalah sekitar 5.3 milyar dengan laju pertambahan penduduk sekitar 2% per tahun. Maka perkiraan penduduk dunia pada awal tahun 2015 adalah sebagai berikut:


```
genius> y0 = 5.3;  # jumlah penduduk pada awal th 1990 (milyar) genius> k = 0.02;  # laju pertumbuhan pertahun genius> t = 2015-1990;  # selang waktu sejak awal th 1990 (tahun) genius> y = y0*exp(k*t)  # jumlah penduduk pada awal th 2015 (milyar) = 8.73822273471
```

Jadi jumlah penduduk dunia pada awal tahun 2015 adalah sebanyak 8.73 milyar.



Contoh 4.

Berikut ini sebuah rangkaian listrik dengan sumber tegangan $v_s = 10\cos(4t)$, $\omega = 4$.



Gambar 2

Anggap Vs adalah notasi fasor untuk sumber tegangan.

Impedansi pada rangkaian tersebut dapat dihitung dengan rumus sebagai berikut

$$Z = R + \frac{1}{i\omega C}$$

Dengan menggunakan rumus tersebut dan data-data yang terdapat pada gambar di atas maka fasor untuk impendansi Z dapat dihitung sebagai berikut:

```
genius> Vs = 10;  # Fasor untuk sumber tegangan
genius> omega = 4;
genius> R = 5;  # Tahanan (ohm)
genius> C = 0.1;  # Kapasitor (F)
genius> Z = R + 1/(1i*omega*C)
= 5 0-2 5i
```

Selanjutnya, arus yang mengalir dapat dihitung dengan rumus sebagai berikut:

$$I = \frac{V_s}{Z}$$

Menggunakan formula ini, arus yang mengalir pada rangkaian dapat dihitung sebagai berikut:

```
genius> Is = Vs/Z
= 1.6+0.8i
genius> Ims = abs(Is)
= 1.788854382
genius> theta = atan(Im(Is)/Re(Is))*180/pi
= 26.5650511771
```


Jadi arus yang mengalir pada rangkaian adalah $i(t) = 1.789 \cos(4t + 26.57^{\circ}) A$.

Contoh 5.

Sebuah beban yang bekerja di atas sebuah balok mempunyai intensitas beban seperti yang ditunjukkan pada gambar di bawah. Intensitas beban tersebut dihampiri dengan fungsi sebagai berikut:

$$w = \sqrt{c_1 x + \sqrt{c_2 + x^2}} \text{ N/m}$$





Gambar 3

Diketahui : $c_1 = 5$ m; $c_2 = 16$ m; a = 3 m; b = 1 m.

Resultan beban (F) yang bekerja pada balok dan letaknya (d) dapat dihitung dengan formula sebagai berikut

$$F = \int_{0}^{a+b} w(x) dx$$

$$d = \frac{\int\limits_{0}^{a+b} xw(x) dx}{\int\limits_{0}^{a+b} w(x) dx}$$

Nilai integral tertentu dari fungsi fct dengan batas integral xl dan xr dapat dihitung dengan fungsi NumericalIntegral(fct,xl,xr).

Berikut ini perhitungan resultan beban (F) dan letaknya (d).

```
genius> function w(x) = sqrt(5*x + sqrt(16 + x^2));
genius> F = NumericalIntegral(w,0,4)
= 14.8851624826
genius> function wx(x) = x*w(x);
genius> M = NumericalIntegral(wx,0,4)
= 33.7350220467
genius> d = M/F
= 2.2663522878
```

Jadi resultan gaya dari beban yang bekerja adalah 14.885 N yang terletak 2.266 m dari kiri balok.

Contoh 6.

Gambar 4 adalah sebuah rangkain listrik sederhana. Penentuan arus I_1 , I_2 dan I_3 yang mengalir pada rangkaian tersebut dapat ditentukan dengan hukum Kirchhoff.

Hukum Kirchoff tentang arus menyatakan bahwa jumlah arus yang masuk pada suatu titik percabangan adalah sama dengan jumlah arus yang keluar dari titik percabangan tersebut. Berikut ini persamaan didapatkan dengan menerapakan hukum tersebut.


$$i_1 + i_2 = i_3$$

Selanjutnya, hukum Kirchoff tentang tegangan menyatakan bahwa jumlah total tegangan yang terdapat pada suatu rangkaian listrik yang tertutup adalah sama dengan nol. Dengan menerapkan hukum ini kita dapat mendapatkan persamaan-persamaan sebagai berikut:

$$10 - 6i_1 - 2i_3 = 0$$

$$-14 + 6i_1 - 10 - 4i_2 = 0$$





Gambar 4

Dari persamaan-persamaan di atas diperoleh persamaan linier sebagai berikut:

$$i_1 + i_2 - i_3 = 0$$

 $6i_1 + 2i_3 = 10$
 $6i_1 - 4i_2 = 24$

Anggap A merupakan matrik koefisien dari persamaan linier dan b adalah vektor yang menyatakan koefisien pada sisi kanan persamaan.

```
genius> A = [1,1,-1; 6,0,2; 6,-4,0]
=
[1.0, 1.0, -1.0
6.0, 0.0, 2.0
6.0, -4.0, 0.0]
genius> b = [0; 10; 24]
=
[ 0.0
10.0
24.0]
```

Arus yang mengalir pada rangkaian dapat dihitung sebagai berikut:

```
genius> i = SolveLinearSystem(A,b)
=
[ 2.0
    -3.0
    -1.0]
```

Diperoleh arus yang mengalir pada setiap tahanan yaitu $i_1 = 2$, $i_2 = -3$, $i_3 = -1$. (Tanda negatif berarti arah arus berlawanan dengan arah yang diasumsikan pada gambar).

Contoh 7.

Sebuah kaleng soda diletakkan di dalam sebuah kulkas. Anggap temperatur kaleng soda adalah $25\,^{\circ}$ C dan temperatur ruangan di dalam kulkas adalah $10\,^{\circ}$ C serta koefisien konduksi dari kaleng soda adalah 0.05.

Penurunan temperatur kaleng soda dalam kulkas dapat dimodelkan dengan persamaan diferensial sebagai berikut:

$$\frac{dT}{dt} = -k(T - T_a)$$


Persamaan diferensial orde pertama dengan nilai awal dapat diselesaikan dengan menggunakan fungsi RungeKuttaFull(fcn,x0,y0,x1,npts). Di mana fcn adalah fungsi yang menyatakan persamaan diferensial, y0 adalah nilai awal, x0 dan x1 adalah rentang waktu penyelesaian, serta npts adalah jumlah titik yang dievaluasi dalam rentang.



Berikut ini perintah-perintah untuk menyelesaikan persamaan diferensial di atas.

```
genius> function Tdot(t,T) = -0.05*(T-10);
genius> t_range = [0,100];
genius> npoints = 500;
genius> T0 = 25;
genius> pt = RungeKuttaFull(Tdot,t_range@(1),T0,t_range@(2),npoints);
genius> LinePlotDrawLine(pt@(,[1,2]),"color","blue","legend","Temperature");
```

Fungsi fTemp adalah fungsi yang merepresentasikan persamaan diferensial, variabel T0 adalah nilai awal kemudan t merupakan rentang waktu penyelesaian.



Penurunan temperatur kaleng soda dalam kulkas adalah seperti yang diperlihat pada gambar 5. Dari gambar tersebut terlihat bahwa setelah menit ke 100 maka temperatur kaleng soda akan sama dengan temperatur kulkas.